
Abstract

In ASP, programs can be viewed as specifications of finite
Herbrand structures. Other logics can be (and, in fact, were)
used towards the same end and can be taken as the basis of
declarative programming systems of similar functionality as
ASP. We discuss here one such logic, the logic FO(ID), and
its implementation IDP3. The choice is motivated by no-
table similarities between ASP and FO(ID), even if both ap-
proaches trace back to different origins.

Introduction
Answer-set programming (ASP), described in other articles
in the issue, is based on an extension of the language of logic
programming under the answer-set semantics. In ASP, an in-
stance of a computational problem is encoded by an answer
set program in such a way that the Herbrand models of the
program determine all solutions for the problem instance.
Thus, at an abstract level, answer set programs are specifi-
cations of finite Herbrand structures (those that are models
of the programs), and the key reasoning task supported by
ASP systems is to compute them. That task is often referred
to as model generation.

Other logics that can express constraints on Herbrand
structures (or even non-Herbrand structures) could also be
used as the basis for this form of declarative problem solv-
ing. Implementing the model generation task for theories in
such logics yields declarative programming tools with the
same basic functionality as that of ASP systems. Two ex-
amples of such an alternative approach are the model gen-
erators NP-SPEC (Cadoli et al. 2000) and aspps (East and
Truszczynski 2006) for extended DATALOG. In both cases,
the specification language is an extension of DATALOG
with clausal constraint rules. This formalism captures the
class NP in the sense that any decision problem from the
class NP can be specified. The system NP-SPEC offers a
compiler for recursion- and negation-free NP-SPEC prob-
lems to SAT whereas aspps is a native system supporting
recursive DATALOG, making it suitable to model problems
involving transitive closure.

A further step in this direction is the logic FO(ID). The
origin of the logic FO(ID) can be traced to the fundamental

problem of logic programming that also triggered the devel-
opment of the stable semantics and later ASP, namely, the
problem of negation as failure (cf. the paper Answer Sets
and the Language of Answer Set Programming (Lifschitz
2016) in this issue). The essence of the negation problem
was that negation as failure in Prolog had so many useful
and intuitive applications, and yet, the original view of logic
programs as sets of material implications (Horn clauses ex-
tended with negation in the body) could not account for the
derivation of a single negative literal. As noted by Lifschitz
in his paper in this issue, one solution was inspired by re-
search in nonmonotonic reasoning and consisted of adapting
the semantics of default logic (Reiter 1980) to the syntax of
logic programs. This led to the stable semantics and later to
ASP.

An alternative solution was to interpret a logic program as
a definition of its predicates. In this view, a logic program
consists of, for every predicate, an exhaustive list of rules
defining the cases in which it is true. Unlike sets of impli-
cations, definitions entail negative information. Moreover,
a case-based representation of definitions and, in particular,
inductive definitions is a common way to specify definitions
in mathematical texts. The view of logic programs as a defi-
nition was already implicit in Clark’s first order completion
semantics (Clark 1978). However, as was well-known in
mathematical logic and databases, in general inductive def-
initions cannot be expressed in first order logic (FO) (Aho
and Ullman 1979). Hence, Clark’s semantics did not not cor-
rectly interpret recursive logic programs as inductive defini-
tions. This weakness spurred the development of so-called
canonical semantics such as the perfect model (Apt, Blair,
and Walker 1988) and the well-founded model semantics
(Van Gelder, Ross, and Schlipf 1991). The latter (in a suit-
ably extended version) turned out to correctly formalize def-
initions, including the most common forms of inductive def-
initions (Denecker 1998; Denecker and Vennekens 2014).
Since definitional knowledge is an important form of expert
knowledge and since in general it cannot be expressed in
FO, it is natural to seek extensions of classical logic to in-
corporate it. This has been recognized by the database com-
munity that developed such extensions in support of more
expressive database query languages (Abiteboul, Hull, and
Vianu 1995). It has also motivated a similar effort in knowl-
edge representation. In particular, these considerations led



to the logic FO(ID), an extension of FO with inductive defi-
nitions expressed as sets of logic program-like rules under
well-founded semantics that represent the base cases and
(possibly) inductive cases of the definition (Denecker 2000;
Denecker and Ternovska 2008).

The logic FO(ID) has a standard Tarskian model seman-
tics. A structure satisfies a theory if it satisfies its FO sen-
tences and is a well-founded model of its definitions. Thus,
the logic FO(ID) can be understood as a conjunction of its
formulas and definitions and, in particular, it is a conserva-
tive extension of first-order logic. These features make the
logic FO(ID) well suited as the basis of declarative knowl-
edge representation systems.

Two systems were developed for significant fragments of
the FO(ID) language: enfragmo (Aavani et al. 2012), and
IDP3 (De Cat et al. 2014). The key inference task sup-
ported by these systems is (finite) model expansion: given
an FO(ID) theory and a partial structure as input, the goal is
to output one or more structures that are models of the theory
and expand the input structure. The enfragmo system pro-
vides support for arithmetic and aggregates in its theories,
and also some limited support for inductive definitions. The
IDP3 system has a similar functionality as enfragmo but pro-
vides a more advanced treatment and support for inductive
definitions; in addition to model expansion, it also supports
several other forms of inference.

In this paper, we present the logic FO(ID) and the IDP3
system. Our presentation is not formal but relies on a se-
ries of examples. We start by illustrating the importance of
structures in knowledge representation and how they lead to
a methodology for knowledge modeling and for declarative
problem solving through the model expansion task. We then
describe the IDP3 system. This discussion is intertwined
with references to the logic FO(ID), the theoretical founda-
tion for the IDP3 system. Next we briefly discuss the re-
lationship between FO(ID) (IDP3) and ASP and conclude
with comments on the role these formalisms may play in the
future of computational logic.

Structures in Knowledge Representation
First-order logic has proved to be a powerful formalism for
representing knowledge largely because of two key interre-
lated factors. Structures used as interpretations of first-order
formulas are well suited to model practical situations and
application domains; and our intuitive understanding of how
first order formulas constrain the space of possible structures
matches exactly the formal definition of the satisfiability re-
lation. Our goal in this section is to present structures in
their role as a fundamental abstraction for knowledge repre-
sentation.

When modeling a problem domain, we start by selecting
symbols to denote its functions and relations. Collectively,
these symbols form the vocabulary of the domain. Each
symbol in the vocabulary comes with a non-negative inte-
ger called the arity that denotes the number of arguments
of the corresponding function or relation. If a is a relation
or function symbol, we write a/k to indicate the arity of a.
Function symbols of arity 0 are called constants.

To illustrate, let us consider a hypothetical software
company that holds weekly lunch meetings for its soft-
ware development teams. These lunch meetings take place
on certain weekdays. Some teams are in a (scheduling)
conflict (for instance, they may share a team member). The
meeting days for teams in conflict must be different.

This short text describes a problem domain. The under-
lined terms indicate relations and functions in that domain.
Figure 1 shows symbols that could be selected to denote
them, as well as their intended meaning. Together, these
symbols form the vocabulary of the lunch meeting domain.

Relation symbols
team/1, day/1, conflict/2

team(x): x is a team
day(x): x is a weekday
conflict(x, y): x has a conflict with y

Function symbols (here only one)
mtng day/1

mtng day(x) = y: the time of lunch meeting of x is y

Figure 1: A vocabulary of symbols for the lunch meeting
domain.

A structure (also called an interpretation) S over a vocab-
ulary consists of a non-empty universeD and, for each sym-
bol σ in the vocabulary, the value σS of σ in S (also called
the interpretation of σ in S). More specifically, for every re-
lation symbol r/k, rS is a relation on D with k arguments
(that is, rS ⊆ Dk), and for every function symbol f/k, fS
is a function onD with k arguments (that is, fS : Dk → D).
Figure 2 shows an example of a structure over the vocabu-
lary from Figure 1.

Domain of S
DS = {T1, T2, T3, T4, T5,M, Tu,W, Th, F}
Relations
teamS = {T1, T2, T3, T4, T5}
dayS = {M,Tu,W, Th, F}
conflictS =
{(T1, T2), (T1, T5), (T2, T3), (T2, T5), (T3, T4)}

Functions
mtng dayS =
{T1 →M,T2 →W,T3 →M,T4 → Tu, T5 → Tu,
M →M,Tu→M,W →M,Th→M,F →M}

Figure 2: A structure S for the vocabulary from Figure 1.

A structure over a vocabulary is an abstract representation
of a concrete instance, or state of affairs, of a problem do-
main modeled in terms of this vocabulary. The universe of
the structure is an abstraction of the set of objects in that in-
stance, while the relations and functions of the structure —
the values of the symbols in the vocabulary — abstract the
relations and functions in the instance.

For instance, the structure S in Figure 2 represents a state
of affairs with five teams referred to as T1, . . . , T5 and five



(working) days referred to as M, . . . , F . The structure also
specifies conflicts between the teams (e.g., teams T1 and T2
are in conflict), and an assignment of meeting days to teams
(e.g., team T3 meets on Monday). Since functions in struc-
tures are total, the function mtng dayS is also defined on
days. This is redundant, as those assignments do not rep-
resent any pertinent information. The structure S is an ab-
straction of a possible state of affairs of our problem domain:
one in which the properties of the domain mentioned in the
specification of this problem domain are satisfied. Indeed,
the mapping mtng dayS assigns weekdays to teams (other
assignments it makes are immaterial or, as we said, redun-
dant), and two teams in conflict are not scheduled to have
lunch on the same day.

To formally model this domain, these and other proper-
ties present in the informal description (some only implic-
itly) need to be expressed as sentences of the logic over the
chosen vocabulary. Three properties relevant to our scenario
are shown in Figure 3. We see there formal sentences in
the language of first-order logic over the vocabulary from
Figure 1, as well as their informal reading. The first sen-
tence says that the relation conflict applies to teams only. It
specifies the types of the arguments of the relation conflict .
Incidentally, this information is not explicitly present in the
narrative. Nevertheless, it is implicit there and can be in-
cluded in any formal representation of the problem. The
second sentence is of a similar nature. It describes the type
of objects the function mtng day maps to. The last sentence
represents the essential constraint of the problem that teams
in conflict do not hold their lunch meetings on the same day.

Vocabulary: As in Figure 1
Sentence:
∀X∀Y (conflict(X,Y )→ team(X) ∧ team(Y ))

Reads: for all X and Y , if X and Y are in conflict then both
X and Y are teams

Sentence: ∀X∀Y (mtng day(X) = Y → day(Y ))

Reads: for all X and Y , if Y is the meeting day for X then Y
is a day

Sentence: ∀X∀Y (conflict(X,Y )→
mtng day(X) 6= mtng day(Y ))

Reads: for all X and Y , if X and Y are in conflict then the
meeting day for X is different from the meeting day for Y

Figure 3: Relevant properties as first-order sentences.

First-order propositions (FO sentences) are true or false in
structures (we also say satisfied or unsatisfied, respectively).
For example, the third proposition of Figure 3 expressing
that conflicting teams do not meet on the same day is true
in the structure from Figure 2. However, it is false in the
structure that is the same except that the interpretation of
conflict is extended with the pair (T4, T5). Indeed, we now
have two teams in conflict that are scheduled to meet on the
same day.

More formally, given a structure S interpreting all sym-
bols in a first-order sentence F , we can evaluate F in S, that

is, assign to it a logical value true or false . When F eval-
uates to true in S, we say that S is a model of F or that F
is satisfied by S . The property of a sentence being true in
a structure yields a satisfaction relation between structures
and first-order logic sentences. It provides first-order logic
sentences with a semantics (the first-order semantics) that
captures precisely their informal reading. Figure 4 illustrates
these concepts for sentences from the language based on our
example vocabulary. The second sentence would evaluate to
false in case the relation conflictS contained the extra pair
(T4, T5).

Vocabulary: As in Figure 1
Structure: S defined in Figure 2
Sentence: ∀X∀Y (conflict(X,Y )→ team(X) ∧ team(Y ))
Evaluates to true . Indeed, for every (a, b) ∈ conflictS , both
team(a) and team(b) hold in S (cf. Figure 2)

Sentence: ∀X∀Y (conflict(X,Y )→
mtng day(X) 6= mtng day(Y ))

Evaluates to true . Indeed, for every (a, b) ∈ conflictS ,
teams a and b meet on a different day (cf. Figure 2)

Figure 4: The first-order semantics applied to some sen-
tences from Figure 3.

The satisfaction relation is of crucial importance. In some
cases, we fully know the relevant state of affairs of the prob-
lem domain or, more precisely, we know the structure that
serves as its abstract representation. However, in other cases
the precise state of affairs is not known or is only partially
known. In that case, our knowledge frequently consists of
separate informal propositions. They implicitly specify pos-
sible states of affairs as those in which these propositions
hold true. As we argued, structures are formal represen-
tations of states of affairs. Those structures that represent
possible states of affairs are called intended. Given this
terminology, knowledge representation can then be under-
stood as the art and practice of formulating knowledge as
a formal theory so that models of that theory are precisely
the intended structures. For instance, the original specifi-
cations of the problem domain are correctly expressed by
axioms in Figure 3, which also shows their informal seman-
tics. The structure from Figure 2 is an intended one, and it
indeed satisfies (is a model of) all the sentences in Figure 3.
That latter claim can be verified formally and is also easy
to see intuitively: conflicts are between teams, the mapping
mtng dayS assigns days to teams (and also to days, but that
is immaterial), and two teams in conflict are not scheduled
on the same day.

To say that all models of the theory are intended structures
here is slightly imprecise. For example, the theory has infi-
nite models which hardly count as intended structures. The
problem is that some implicit information such as what are
weekdays and teams, is not expressed in the theory. This in-
formation is expressed in the values assigned to the symbols
team and day by structures like the one in Figure 2.

To recapitulate, in this setting the satisfaction relation al-
lows us to use sentences over the fixed vocabulary to con-



strain structures over that vocabulary to those that satisfy
the sentences. These sentences can be seen as specifications
of classes of intended structures over that vocabulary, that
is, the structures that represent those states of affairs that are
possible (might be encountered in practice).

The setting we presented supports several important rea-
soning problems. Say the manager in our running example
is reviewing a schedule proposed by one of her assistants
or, more formally, the corresponding structure. The man-
ager wants to know whether certain propositions hold for
the schedule or, formally, whether the formal sentences ex-
pressing the propositions are satisfied in that structure. We
call that reasoning task model checking or querying. For
instance, we might want to know whether team T2 has its
meeting scheduled on the same day as team T4 in the struc-
ture (schedule) in Figure 2 (that query would evaluate to
false). Model checking is a special instance of this task;
it verifies that a structure satisfies the specifications, that is,
that it indeed is an intended structure. In the case of our ex-
ample and the structure in Figure 2, it consists of verifying
that all statements of the theory in Figure 3 are satisfied by
the structure.

Even more interesting and important is the situation when
the schedule is yet to be constructed. How can the manager
find one? She knows the axioms in Figure 3. This informa-
tion specifies the class of intended structures, each of them
representing a valid instance of a lunch meeting domain. It
is also reasonable to assume that she knows which teams she
needs to schedule, what scheduling conflicts she has to take
into account, and which days are work days. This informa-
tion explicitly fixes the domain of an intended structure as
well as its relations team , day and conflict (for instance, to
the values they have in Figure 2). Any function mtng day
that completes this explicitly given fragment of a structure
to an intended one yields a good schedule for the setting of
interest to the manager. The converse is also true. Good
schedules give rise to intended structures (when combined
with the explicitly given components).

The task to find the missing function, which we just de-
scribed, is an example of the model expansion problem. In
model expansion we assume that the vocabulary is parti-
tioned into input and output symbols. Given a theory (that is,
a set of sentences) over the entire vocabulary and a structure
over the vocabulary consisting of the input symbols, called
an input structure, the goal is to extend the input structure
with relations and functions for output symbols so that the
resulting structure (now over the entire vocabulary) satisfies
the theory.

This applies to our example scenario. Here, team , day
and conflict are input symbols and mtng day is an out-
put symbol. The input structure consists of the domain
{T1, . . . , T5,M, . . . , F} and of the relations team , day , and
conflict as in Figure 2. The theory specifying intended
structures is given in Figure 3. Under these assumptions,
the model expansion problem asks for a specific function
mtng day that would expand the input structure to the one
modeling the three sentences (that is, to an intended struc-
ture). That function would offer a legal schedule of lunch
meetings for the five teams involved. The function shown in

Figure 2 is one of possible solutions. The function in Figure
5 is another one.1

mtng day =
{T1 →M,T2 →W,T3 →M,T4 →W,T5 → Tu,
M →M,Tu→M,W →M,Th→M,F →M}

Figure 5: Another possible schedule function.

A more involved reasoning task is assigning meeting days
to teams interactively. The task involves propagation infer-
ence that calculates the valid days for every team. Each time
the user assigns a meeting day to a team, the propagation
inference updates the valid choices for the remaining teams.
Still another task is revision, assisting a user in assigning a
different meeting day to a particular team while preserving
as much as possible the meeting days assigned to the other
teams. All these reasoning tasks use the same theory as a
specification of valid structures.

To recapitulate, structures are important to us for four key
reasons. First, they provide natural abstractions of states of
affairs of the problem domain, in which sentences (prop-
erties) can be evaluated for satisfaction. Second, they are
useful to define computational tasks in the context of logic.
Third, they can be used to present input data, as in the query
inference (where the value of every symbol is known), or
in model expansion (where values of some symbols are not
known). Fourth, they can be used as representations of an-
swers to model expansion problems.

The IDP3 System
We will now present a software system IDP3 that imple-
ments the ideas presented above. In particular, IDP3 allows
us to define structures, input structures and partial structures,
as well as sentences to state their properties.2 An overview
of the IDP3 system is presented in Figure 6. We use a series
of simple examples to illustrate and discuss all key features
of IDP3.

The IDP3 system separates information from the reason-
ing task to be performed. In this way, it facilitates the use
of the same knowledge to solve diverse reasoning problems.
To represent information, IDP3 uses an enriched variant of
first order logic. The information is split over three com-
ponents. The first component is the vocabulary. The IDP3
syntax for describing vocabularies is illustrated in Figure 7,
where we again use our software teams domain as an exam-
ple. The vocabulary goes beyond the basic first order logic

1A more general version of the model expansion problem takes
a partially instantiated structure (a fully specified domain but pos-
sibly only partially instantiated relations and functions for all vo-
cabulary symbols) and asks if it can be completed to a structure
that satisfies the theory.

2The IDP3 system has been developed by the Knowledge
Representation and Reasoning group at the University of Leu-
ven https://dtai.cs.kuleuven.be/topics/kbs. The
most recent versions of the source code and documentation, as well
as other resources such as an on-line IDE are available at the IDP
page https://dtai.cs.kuleuven.be/software/idp.



• Vocabulary

• Structure

• Theory

• Procedure

• Term

• Query

• Model checking
(query inference)

• Model expansion

• Propagation

• Revision

• Optimization

• Deduction

Lua code embedded in procedure components for calling
inference methods

Procedural interface: Lua

Inference MethodsLanguage Components

Figure 6: High-level representation of IDP3 as a Knowledge
base System.

as it introduces not only the alphabet but also the types and
the signatures of the relation and function symbols. In IDP3,
vocabularies are assigned identifiers; the vocabulary in Fig-
ure 7 is identified by V .

vocabulary V{
type day
type team
conflict(team,team)
mtng day(team):day

}

Figure 7: An IDP3 vocabulary introducing two types day
and team, and a predicate and a function together with their
signatures.

The second component is a structure. The IDP3 syntax
for a structure for our example domain is shown in Fig-
ure 8. The structure has an identifier (here S). It refers to
a vocabulary (here V ) and introduces the domains for the
types declared in the vocabulary (here by enumerating the
elements of types day and team). In addition it also enu-
merates the known information about the declared relations
and functions. The interpretation of conflict is fully known,
and the relation is specified by the list of its tuples. On the
other hand, as nothing is known about the interpretation of
mtng day , nothing about this function is mentioned in the
structure.

structure S:V{
day={M;Tu;W;Th;F}
team={T1;T2;T3;T4;T5}
conflict={(T1,T2);(T1,T5);(T2,T3);(T2,T5);(T3,T4)}

}

Figure 8: An IDP3 structure for the vocabulary V from Fig-
ure 7.

The last component used to express information is a the-
ory. An example theory appropriate for our domain is shown

in Figure 10. The notation is essentially first order logic
but in a keyboard-friendly syntax. The IDP3 counterparts to
standard mathematical logic notation are given in Figure 9.

FO IDP3 FO IDP3 FO IDP3
∧ & ≡ <=> = =
∨ | ¬ ∼ 6= ∼=
→ => ∀ ! ≤ =<
← <= ∃ ? ≥ >=

Figure 9: Translation table. Note that implication translates
into a double arrow, while the translation of ≤ and ≥ is not
an arrow.

The theory (as the other two components) has an identi-
fier (here T ). The theory refers to a vocabulary (in the ex-
ample, via the identifier V ), and expresses the constraints
that a structure must meet to serve as a valid abstraction of
the problem domain (here, the function mtng day is con-
strained so that teams in conflict are not scheduled for their
lunch meetings on the same day). The types of the variables
are optional; if omitted, type inference will derive them. In
the case at hand, the type of the variables a and b can be
derived from the signature of conflict and mtng day .

theory T:V{
! a[team] b[team]: conflict(a,b) =>

mtng day(a) ∼= mtng day(b).
}

Figure 10: An IDP3 theory over the vocabulary of Figure 7
(using correspondences from Figure 9).

For solving problems, the IDP3 system offers a procedu-
ral interface in Lua3 and executes the procedure main(). In
main(), the IDP3 user can overwrite default values of solver
options, can invoke Lua functions provided by the IDP3 de-
signers as well as standard Lua functions, and she can also
invoke functions written by herself in the procedural compo-
nent. The information components (vocabularies, structures
and theories) are first-class citizens in the Lua code and can
be passed as parameters to various functions. The proce-
dural component in Figure 11 illustrates some common use
patterns of the Lua interface. In this code, modelexpand,
printmodels and allmodels are Lua functions provided
with the IDP3 system, while # is the Lua operator return-
ing the length of a Lua sequence and print is the standard
Lua printing function.

The modelexpand function invokes model expansion in-
ference on a theory T and a structure S (both referring
to the same vocabulary), and returns a (possibly empty)
Lua sequence of models of T that extend S; by default,
modelexpand is bound to return a single model. To obtain
more models, another bound can be set with an assignment
to the stdoptions.nbmodels option. The printmodels

3Lua is a scripting language (Ierusalimschy, Henrique de
Figueiredo, and Celes 1996) available at http://www.lua.
org.



procedure main(){
stdoptions.nbmodels=5
printmodels(modelexpand(T,S))
models = allmodels(T,S)
print(#models)
print(model[5])
print(model[980])

}

This procedure applies model expansion inference on our the-
ory T and initial structure S in two different ways. The first
line sets the bound on the number of models to 5. The sec-
ond line invokes model expansion and prints the sequence of
5 models. The third line also invokes model expansion but re-
turns the sequence of all models and assigns it to the Lua vari-
able models. The fourth line prints the number of models in
the sequence, the next line prints the fifth model and the final
line prints nil as there are only 960 models.

Figure 11: Solving the model expansion problem in IDP3.

function prints the number of models in a sequence of mod-
els as well as each of its models. To obtain the sequence
of all models, one can use the allmodels function. Indexing
can be used to select a particular element in a sequence; if
the sequence is empty (models do not exist), the special Lua
value nil is returned. Models are represented and printed as
structure components so that they can serve as IDP3 input.

Definitions, Aggregates and Optimization
So far, we have seen two extensions of first order logic that
are available in IDP3: types and partial functions (a typed
function is partial as it is only defined for the values deter-
mined by the types in the signature). Other important ex-
tensions are aggregates and definitions. To illustrate them,
we elaborate on our example; at the same time we also in-
troduce another reasoning task, optimization inference. The
IDP3 code for the extended example is shown in Figure 12.

In the extended scenario, we are concerned with the work-
load of the company cafeteria where the meetings take place.
We introduce the concept of quiet day that we define as a
day in which at most one team holds its meeting. As stated
before, definition expressions in FO(ID) are modeled after
the way definitions are expressed in text. They define one or
more predicate or function symbol in terms of a set of pa-
rameter symbols. E.g., the concept quiet day is defined in
terms of the function mtng day which we call a parameter
of the definition. To distinguish a definition expression from
FO sentences, it is written as a set of rules placed between
“define {” and “}”4.

Each rule expresses one (base or inductive) case. Head
and body of the rule are separated by “<-”, called the def-
initional implication to distinguish it from the material im-
plication => (both given in the IDP syntax). The head is
an atomic formula of one of the defined predicates and the
body can be any formula in first order logic. In contrast
with logic programming, variables are explicitly quantified.

4The keyword define is optional.

vocabulary V{
...
type number isa nat
quiet day(day)
nmbr mtngs(day):number

}
structure S:V{

...
number={0..10}

}
theory T:V{

...
define {

! d: quiet day(d) <- ∼(? t1 t2: t1 ∼= t2 &
mtng day(t1)=mtng day(t2)=d).

}
define {

! d: nmbr mtngs(d)=#{tm: mtng day(tm)=d}.
}

}
term m:V{

max{d[day] : true : nmbr mtngs(d)}
}
procedure main(){

stdoptions.cpsupport=true
models, optimal, cost = minimize(T,S,m)
print(models[1])
print(optimal)
print(cost)

}

Figure 12: The running example extended to illustrate ag-
gregates, definitions and optimization. The missing lines of
code should be taken from Figures 7, 8, and 10.

To give such formal rule sets the intuitive reading of def-
initions in mathematics, the semantics chosen for them is
an extension of the well-founded semantics (Van Gelder,
Ross, and Schlipf 1991; Denecker and Ternovska 2008), be-
cause the well-founded semantics correctly formalizes the
most common forms of definitions found in text (Denecker
and Vennekens 2014). The IDP3 definition given in Figure
12 formally expresses the intended meaning for the concept
quiet day : a day d is a quiet day if it is not the case that two
different teams (t1, t2) have their meeting day on d.

Similarly, we include in the vocabulary a function
nmbr mtngs that we want to define as the function that
maps a day to the number of teams meeting on this day. The
new symbol ranges over the new type number , which we
introduce in the vocabulary as a subtype of the natural num-
bers (a built-in type nat) and specify in the structure as the
set of numbers from 0 to 10. As the relation quiet day , also
this function is defined in terms of the parameter mtng day .
In general, functions are defined by sets of rules of the
form f(t1, . . . , tn) = t <- body . In the case at hand, the
body degenerates to “true” and is omitted. Importantly, the
function value here is given by the cardinality aggregate
#{tm : mtng day(tm) = d}. This aggregate represents
the cardinality of the set {(tm) | mtng day(tm) = d}, i.e.,
the number of teams meeting on day d. Besides cardinal-



ity, IDP3 also supports minimum, maximum, sum and prod-
uct aggregates. They have a slightly different syntax. An
overview of the supported aggregates is given in Figure 13.

Assume that the manager of the cafeteria wishes to min-
imize the maximal workload for the cafeteria. To solve
this problem, another form of inference is needed called
optimization inference. This is done by the procedure call
minimize(T, S,m) in the main() procedure of Figure 12.
The procedure call contains yet another sort of component
of IDP3: the term component. Its role is to give a name
to a term. Referred to by its name, the term can then
be used inside Lua procedures. Here, the term of inter-
est is the maximum n in the set of pairs (d, n) defined as
{(d, n) | n = nmbr mtngs(d)}. This is a simple maxi-
mum aggregate in which there are no extra conditions on d.
According to the translation table of Figure 13, its IDP3 syn-
tax is max{d [day ] : true : nmbr mtngs(d)}. The mid-
dle part “true” is the trivially true extra condition on the
selected values for d. The optimization inference performs
a search for a model that minimizes the value of the term
referred to by m. The call not only returns a model, but also
whether optimality could be shown and the value of the term.
So, minimization is on the maximal number of meetings on
the same day. In other words, the call minimize(T, S,m)
returns a schedule that minimizes the maximum number of
lunch meetings scheduled for a single day (informally, it of-
fers a “balanced” schedule). The grounder of the IDP3 sys-
tem is unable to derive a bound on the value of the opti-
mization term m. To avoid an infinite grounding, the option
cpsupport must be on.

FO IDP3

#{x : F} #{x1 ... xn : F}
sum{(x, t) : F} sum{x1 ... xn : F : t}
prod{(x, t) : F} prod{x1 ... xn : F : t}
max{(x, t) : F} max{x1 ... xn : F : t}
min{(x, t) : F} min{x1 ... xn : F : t}

Figure 13: Translation table for aggregates.

Partial Information and Constructed Types
As an alternative elaboration of our example, assume it is
decided that team T1 meets on Monday (“is certainly true”)
and team T2 does not meet on Tuesday (“is certainly false”).
This partial knowledge can be expressed in the structure as
shown in Figure 14 (we note the use of markers <ct> and
<cf>).

structure S:V{
...
mtng day<ct> = {T1 ->M}
mtng day<cf> = {T2 ->Tu}

}

Figure 14: Partial knowledge in a structure.

Alternatively, we may want to express this information

in the theory. However, in the theory we can only express
information about domain elements if we have symbols in
the vocabulary to refer to them. Hence, we need to ex-
tend the vocabulary with constantsmon, tue, . . . , t1, t2, . . .
to denote days and teams; furthermore, the structure needs
to be extended to specify the interpretation for the new con-
stants by means of the statements mon = M, . . . , t1 =
T1, . . . . Only then we can express constraints such as
mtng day(t1 ) = mon or mtng day(t2 )∼=tue in the the-
ory. This verbose way is a consequence of the fact that func-
tions and constants are not limited to their Herbrand inter-
pretation as in ASP and Prolog. A shortcut is to make use of
constructed types to enforce Herbrand interpretations over
certain types. Figure 15 shows how constructed types im-
pose the same constraints on the function mtng day . As the
domain of these types is fixed in the vocabulary, they are not
part of any structure.

vocabulary V{
type day constructed from {M,Tu,W,Th,F}
type team constructed from {T1,T2,T3,T4,T5}
...

}
theory T:V{

...
mtng day(T1) = M.
mtng day(T2) ∼= Tu.

}

Figure 15: Constructed types.

More about Definitions
The definitions we discussed above are simple and can be
expressed in first order logic as equivalences. For exam-
ple, the equivalence !d : quiet day(d)<=>∼(?t1 t2 :
t1∼=t2 & mtng day(t1) = mtng day(t1) = d) correctly
expresses the definition of quiet day . While this works for
all non-inductive definitions, it is well known that inductive
definitions in general cannot be expressed through FO equiv-
alences.

Definitions are the most substantial extension that IDP3
offers with respect to first order logic. Not only do they offer
the designer a facility to define concepts, they also increase
the expressiveness. The archetypal example of a relation
that cannot be expressed in first order logic is the transitive
closure of the edges in a graph. The inductive definition of
this relation, say T , for a graph (N,E) with nodes N and
edges E is often stated as follows:
• If (a, b) ∈ E, then (a, b) ∈ T ,
• If for some c ∈ N , it holds that (a, c) ∈ T and (c, b) ∈ T ,

then also (a, b) ∈ T .
In IDP3, we can model it as in Figure 16.

To further illustrate the power of definitions, we present
in Figure 17 a representation of a simple graph problem that
requires selecting edges among nodes so that in the resulting
graph all vertices are reachable from a (given) node root and
none of (given) forbidden edges are selected. The main diffi-
culty is that the set of vertices reachable from the root is not



vocabulary V{
type node
edge(node,node)
trans(node,node)

}
structure S:V{

edge={...}
}
theory T:V{

define {
! x y: trans(x,y) <- edge(x,y).
! x y: trans(x,y) <- ? z: trans(x,z) & trans(z,y).
}

}

Figure 16: Transitive closure of the edges in a graph.

expressible in first order logic. To overcome this problem,
we introduce the auxiliary unary predicate symbol reach-
able and express it through the inductive definition provided
in Figure 17. Additional axioms express that the defined
relation reachable is the set of all nodes and no edges are
forbidden. An interesting aspect is that here, the defined re-
lation reachable is known initially while the parameter edge
in terms of which it is defined is unknown. Hence, IDP3
searches for an interpretation of the parameter edge such that
the defined relation reachable has the given value. This sort
of input/output pattern is different from that of Prolog and
DATALOG systems, and it shows the declarative nature of
definitions. It is a powerful aspect of IDP3 as well as ASP
systems.

In the example in Figure 17, one can check that the edges
(A,D) and (D,C) must appear in every solution for the re-
lation edge. Also, at least one of (D,B) or (C,B) must
be present. Other allowed edges are not constrained. Thus,
one possible value for edge is {(A,D), (D,B), (D,C)} and
another one is {(A,D), (C,B), (D,C), (B,D)}.5

FO(ID) (IDP3) and ASP
On the conceptual level, FO(ID) and ASP are quite dif-
ferent. Whereas ASP has its foundation in nonmonotonic
and commonsense reasoning, FO(ID) is based on a defini-
tion construct inspired by the structure of definitions used in
mathematics. Negation in ASP is viewed as a non-classical
epistemic or default operator. In FO(ID), it is the defini-
tional rule operator ← that is non-classical, while nega-
tion in the bodies of definition rules is classical. And yet,
despite these different foundations, there are strong struc-
tural relationships between ASP and FO(ID). On the lan-
guage level, FO(ID)’s rule-based definition construct resem-
bles ASP rules, and FO axioms resemble ASP constraints.
We illustrate these similarities with the problem of finding a
Hamiltonian cycle in a directed graph. An answer set pro-
gram encoding the problem is shown in Figure 18.

That program has a typical structure resulting from

5This theory can be accessed and experimented with on
the IDP-IDE webpage at http://dtai.cs.kuleuven.be/
krr/idp-ide/?present=forbidden.

vocabulary V{
type node
forbidden(node,node)
edge(node,node)
reachable(node)
root:node

}
structure S:V{

node = {A..D} // a shorthand for {A; B; C; D}
forbidden = {(A,A); (A,B); (A,C); (B,A);

(B,B); (B,C); (C,C); (C,D); (D,D)}
root = A

}
theory T:V{

// inductive definition of reachable
define {

reachable(root).
! x: reachable(x) <- ? y: reachable(y) & edge(y,x).
}

// The graph is fully connected
! x: reachable(x).
// No forbidden edges
! x y: edge(x,y) => ∼forbidden(x,y).

}

Figure 17: A graph problem.

generate {In(x, y)}<-Edge(x, y).

define T (x, y)<-In(x, y).
T (x, y)<-T (x, z), T (z, y).

test <-In(x, y), In(x, z), y 6= z.
<-In(x, z), In(y, z), x 6= y.
<-Node(x),Node(y),not T (x, y).

Figure 18: A generate-define-test ASP program encoding
the existence of a Hamiltonian cycle problem.

following the generate-define-test (GDT) methodology
(Lifschitz 2002) (discussed in this issue by Faber, Gebser
and Schaub (2016)). This methodology leads to three sorts
of modules. The first of them generates the space of can-
didate solutions (in our example, the space of all subsets of
the set of edges of the input graph; they are possible instan-
tiations of a relation In). The generate module commonly
relies on the construct of choice rules (as in our example)
or, alternatively, uses disjunctive rules. The second one de-
fines some additional concepts that are useful in identify-
ing solutions (here, the transitive closure of the relation In).
Finally, the third one specifies constraints of the problem.
These constraints narrow down the space of candidate solu-
tions to those that represent the valid ones (here, the con-
straints ensure that exactly one edge comes into each node,
exactly one edge leaves each node and, finally, that all nodes
are connected to each other both ways; that last condition
requires an auxiliary concept of the transitive closure). The
horizontal lines in Figure 18 make this structure explicit.

The corresponding IDP3 solution to the problem has a
similar format. We present its theory component in Figure



19.

theory T:V{
! x y: In(x,y) => Edge(x,y).

define {
! x y: T(x,y) <- In(x,y).
! x y z T(x,y) <- T(x,z) & T(z,y).
}

! x y z: ∼(In(x,y) & In(x,z) & y∼= z).
! x y z: ∼(In(x,z) & In(y,z) & x∼= y).
! x y: ∼(Node(x) & Node(y) & ∼T(x,y)).

}

Figure 19: An IDP3 theory encoding the existence of a
Hamiltonian cycle problem.

The similarity is striking. The first sentence plays the role
of the generate module in the program in Figure 18. The
definition of the transitive closure mirrors the define module.
Finally, the last three sentences are the three constraints of
the test module cast in the IDP3 syntax. As an aside, we
note that a direct translation from natural language to the
IDP3 syntax of these constraints would more likely be as in
Figure 20.

! x y z: In(x,y) & In(x,z) => y = z.
! x y z: In(x,z) & In(y,z) => x = y.
! x y: T(x,y).

Figure 20: A direct IDP3 representation of the test con-
straints.

Almost all GDT programs can be translated into IDP3 fol-
lowing the idea outlined above. The encoding of the gener-
ate module does not require any special syntax. In fact, in
many cases the generate part of a GDT program disappears
entirely from the corresponding IDP3 theory. The converse
is also true. A large class of IDP3 theories allows for au-
tomated rewritings into the language of ASP. The key in
such translations is to properly construct the choice rules to
“open” some of the predicates.

Similarities between ASP and FO(ID) can be found not
only in the structure of programs (theories). On the system
level, the core of IDP3 is a model generator that is developed
using similar technologies as current ASP solvers, and offers
similar functionalities.

Concluding Remarks
In this paper, we focused on the model-generation task be-
cause of its natural applications in solving search and opti-
mization problems. This is also the focus of ASP and ASP
implementations. We noted that model generation can be
implemented for other logics. We mentioned some of them
and then described in detail the logic FO(ID) and the asso-
ciated reasoning system IDP3.

However, it is important to point out that the knowl-
edge present in both FO(ID) theories as well as in answer

set programs can support many other reasoning tasks be-
sides model generation. That observation has played a cen-
tral role in the development of the system IDP3 and is re-
flected in its functionality (cf. Figure 6). Similarly, it un-
derlined some developments in ASP (cf. the paper by Kauf-
mann et al. (2016) in this issue). In particular, most im-
plementations of ASP support skeptical and brave reason-
ing, and add-ons facilitating abduction and planning were
developed for some systems, as well (Eiter et al. 2003;
1999).

The field of computational logic has an urgent need for
integrative frameworks that recognize that many reasoning
tasks are needed in knowledge-intensive applications and
that these tasks can all be driven by a single well-designed
underlying knowledge base. Formalisms and systems dis-
cussed in this special issue are on the intersection of sev-
eral related lines of research, building on the advances in
classical logic, automated reasoning, logic programming,
databases, satisfiability, satisfiability modulo theories, con-
straint programming, fixpoint logics, and description log-
ics. As such, they are well suited to play this integrative
role. Their modeling capabilities which, in important re-
spects such as the ability to capture inductive definitions,
go beyond SAT/CSP formalisms, as well as the computa-
tional effectiveness of their reasoning software demonstrate
that. We posit that developing FO(ID), ASP and related
formalisms with this goal in mind is essential both for the
theory of logic-based computation and for practical applica-
tions.

